

Advent of Code 2024

Literate Clojure solutions to programming puzzles.

Rens Oliemans

December 4, 2024

Contents

1	Day 1	4
1.1	Part 1	4
1.2	Part 2	5
2	Day 2	6
2.1	Part 1	6
2.2	Part 2	7
2.2.1	Benchmark results	8
3	Day 3	9
3.1	Part 1	9
3.2	Part 2	9
4	Day 4	11
4.1	Part 1	11
4.2	Part 2	13

Finding the Chief Historian! This program contains my Advent of Code solutions for 2024, which you can find on my sourcehut and GitHub. I believe GitHub doesn't show the results of code blocks, which means that viewing it there might leave you a bit confused.

In general, I've added line numbers to code blocks when that code block is part of the solution file. In some cases, I've added some code that explains, clarifies, justifies or otherifies something. Those lines aren't numbered if they aren't necessary to the final solution.

Utils

I define some common functions in `aoc.util`, mostly related to parsing the input. The input always comes in a file but also usually has an example input. The former is a file (which we read as a string with `slurp` and the latter is just a string in the same format. Therefore it's easiest to let the days itself take care of reading the file (since they also have the example input), and just operate on strings here.

```
1 (ns aoc.util
2   (:require [clojure.string :as str]))

3 (defn string-as-lines
4   "Outputs the string as a vector, one element per line."
5   [input]
6   (str/split input #"\n"))
```

Often the lines contain numbers:

```
7 (defn string-as-numbers-per-line
8   "Assumes there is a number on each line: we parse it and return a
9   vector, one element per line. Technically each number is parsed with
10  `read-string`, so it isn't just limited to numbers, but I've only
11  tested numbers."
12  [input]
13  (let [lines (string-as-lines input)]
14    (map read-string lines)))
```

or lists of numbers. This was the case in both Day 1 and Day 2, where the input had the following format:

```
7 6 4 2 1
1 2 7 8 9
9 7 6 2 1
1 3 2 4 5
8 6 4 4 1
1 3 6 7 9
```

Here, we want the whole file to be represented by a vector, where each element is itself a vector of the space-separated numbers on a line.

```
15 (defn num-list-per-line
16   "Returns a vector of vectors, the outer vector has an element per
17   line, the inner has space-separated elements. "
18   [input]
19   (let [lines (string-as-lines input)]
```

```
20      (->> lines
21          (map #(str/split % #"\\s+"))
22          (map #(map read-string %))))))
```

This function, run on the example table seen above, will return the following:

```
(num-list-per-line example)
```

```
((7 6 4 2 1) (1 2 7 8 9) (9 7 6 2 1) (1 3 2 4 5) (8 6 4 4 1) (1 3 6 7 9))
```

1 Day 1

1.1 Part 1

We need to reconcile two lists. We get them in the following form:

```
3 4
4 3
2 5
1 3
3 9
3 3
```

And our goal is to find the “distance” between the two lists.

To find the total distance between the left list and the right list, add up the distances between all of the [sorted] pairs you found.

For the example above, the correct answer is **11**.

My strategy is: convert the input to pairs of numbers (`aoc.util/num-list-per-line` takes care of this), transpose them (so we have two lists), sort them, transpose them again (pairs), and take the difference and sum it. Makes sense? We need the two tiny helper functions `sum` and `transpose`:

```
1 (ns aoc.1)

2 (defn- sum "Finds the sum of a vector of numbers" [vec]
3   (reduce + vec))
4
5 (defn- transpose "Transposes a matrix" [m]
6   (apply mapv vector m))
```

With the final function being now quite easy to follow if you keep my strategy above in mind. Recall that the correct answer for the testinput was 11.

```
7 (defn p1 [input]
8   (let [input (aoc.util/num-list-per-line input)]
9     (->> input
10      (transpose)
11      (map sort)
12      (transpose)
13      (map #(abs (- (first %) (second %))))
14      (sum))))
15
16 (p1 testinput)
```

11

It works for the testinput, fantastic. Now let's open the file and run it on the input. The input file for day 1 can be found in the file `inputs/1`.

```
17 (def input (slurp "inputs/1"))
18 (p1 input)
```

2057374

Hurrah! We get a **Gold Star**!

1.2 Part 2

Now, we need to find a “similarity score” for the two lists:

Calculate a total similarity score by adding up each number in the left list after multiplying it by the number of times that number appears in the right list.

A naive way to do this would be to iterate over the first list, where, for each element, we count how many items in the second list are equal to that element, and multiply the element with the count. However, you'd be doing a lot of duplicate counting. A faster way to do it is to convert the second (it doesn't really matter which one you pick) list to a map once, with `{element frequency}`. Let's use the function `frequencies`!

```
(frequencies (last (transpose (aoc.util/num-list-per-line testinput))))
{4 1, 3 3, 5 1, 9 1}
```

Now, we can iterate over the first list (which we get by `(transpose (numbers input))`), multiply the element itself by the count in `frequencies`, and sum the result.

```
19 (defn p2 [input]
20   (let [input (transpose (aoc.util/num-list-per-line input))
21         one (first input)
22         freqs (frequencies (second input))]
23     (->> one
24         (map #(* % (freqs % 0)))
25         (sum))))
26
27 (assert (= 31 (p2 testinput)))
28 (p2 input)
```

23177084

2 Day 2

2.1 Part 1

Analysing some unusual data from a nuclear reactor. The data consists of *reports* separated by lines, each of which is a list of numbers (*levels*), separated by spaces.

```
7 6 4 2 1
1 2 7 8 9
9 7 6 2 1
1 3 2 4 5
8 6 4 4 1
1 3 6 7 9
```

We need to find out how many reports are **safe**, which is the case if all levels are gradually increasing or decreasing. This is defined as such:

[A] report only counts as safe if both of the following are true:

- The levels are either all increasing or all decreasing.
- Any two adjacent levels differ by at least one and at most three.

In the example input, there are **2** safe reports—the first and last.

Let's convert all numbers to the difference between the previous number. Then, a report is safe if all numbers are of the same sign, and the absolute of the number is between 1 and 3.

Since we're computing the difference between each element and the element before, I want to use `partition`, which does exactly this. Then, we can use `mapv` to compute the difference. For the last element of the `testinput`:

```
1 (ns aoc.2)
2 (defn diffs [record]
3   (->> record
4     (partition 2 1)
5     (mapv (fn [[a b]] (- b a)))))
```

test it out:

```
(diffs (last (aoc.util/num-list-per-line testinput)))

[2 3 1 2]
```

Now just use that to determine whether a record is safe. Recall that the `testinput` had **2** safe records.

```

6  (defn is-safe? [record]
7    (let [differences (diffs record)]
8      (and (every? #(<= 1 (abs %) 3) differences)
9           (apply = (map pos? differences)))))
10
11 (defn p1 [input]
12   (->> (aoc.util/num-list-per-line input)
13        (filter is-safe?)
14        (count)))
15
16 (p1 testinput)

```

2

```

17 (def input (slurp "inputs/2"))
18 (p1 input)

```

242

2.2 Part 2

Now, the same rules apply as before, except if removing a single level from an unsafe report would make it safe, the report instead counts as safe.

First I had a smart idea. Check out [e2dcab2f0de76c21477c5e871e029f0282c8fab](#). It is much more efficient than the current solution, but much more convoluted and ugly to read. Right now, I just remove each level one by one and check if the record is safe then.

```

19 (defn drop-nth [coll n]
20   (keep-indexed #(if (not= %1 n) %2) coll))
21
22 (defn dampened-is-safe? [record]
23   (some is-safe? (map #(drop-nth record %)
24                       (range (count record)))))
25
26 (defn p2 [input]
27   (->> (aoc.util/num-list-per-line input)
28        (filter dampened-is-safe?)
29        (count)))
30
31 (p2 testinput)

```

4

32 (p2 input)

311

2.2.1 Benchmark results

The old solution took on average 3.8 milliseconds to execute (p2 input), and the new solution about 6.0. This is worth it, IMO, since the code is *much* simpler. Next time, first do the easy thing, and then benchmark to see if it needs to be improved!

3 Day 3

3.1 Part 1

We have an input string that contains a lot of characters, for example:

```
xmul(2,4)%&mul[3,7]!@~do_not_mul(5,5)+mul(32,64]then(mul(11,8)mul(8,5))
```

The goal is to extract all substrings that are of the exact form `mul(\d+,\d+)`, and in that case multiply the two numbers together. This is straightforward, I'm not really going to create any helper functions: parse with regex, convert to int, multiply and sum.

```
1 (ns aoc.3)

2 (defn p1 [input]
3   (let [matches (re-seq #"mul\((\d+),(\d+)\)" input)]
4     (->> matches
5       (map #(list (Integer/parseInt (nth % 1)) (Integer/parseInt (nth % 2))))
6       (map #(apply * %))
7       (reduce +))))

8 (let [input (slurp "inputs/3")]
9   (p1 input))
```

155955228

3.2 Part 2

We get a new example string for Part Two:

```
xmul(2,4)&mul[3,7]!~don't()_mul(5,5)+mul(32,64](mul(11,8)undo()?mul(8,5))
```

This contains the substrings `don't()` and `do()`, which disable and enable `mul()` instructions. I can do fancy clojure things, but Emacs is way too good for this, so let's do it quickly in Emacs. We want to remove everything from the input file that's in between a `don't()` and a `do()` instruction, and then call `(p1)` on this input. There are three slightly tricky things about this:

- The input file has some newlines, and in some cases a `do()` instruction is on a later line than the previous `don't()` instruction.
- You need to match non-greedy in between a `don't()` and a `do()`.
- If you call `(replace-regexp)` with just the regex and replacement string, it will move point to the last match. This is easily fixed by adding the fourth and fifth arguments to `replace-regexp`: `START` and `END`.

So, here's some elisp code that does that.

```
(with-temp-buffer
  (insert-file-contents "inputs/3")
  (replace-regexp "\n" "" nil (point-min) (point-max))
  (replace-regexp "don't().+?do()" "" nil (point-min) (point-max))
  (write-region (point-min) (point-max) "inputs/3-enabled"))
```

And back to clojure for the now trivial second part.

```
10 (let [fixed-input (slurp "inputs/3-enabled")]
11   (p1 fixed-input))
```

100189366

4 Day 4

4.1 Part 1

We need to find all instances of **XMAS**, appearing in a text like below, either horizontally, vertically, or diagonally, including written backwards. According to these rules, the example below contains **18 XMAS**-es.

```
MMMSXXMASM
MSAMXMSMSA
AMXSXMAAMM
MSAMASMSMX
XMASAMXAMM
XXAMMXAMA
SMSMSASXSS
SAXAMASAAA
MAMMMXMMM
MXMXAXMASX
```

My idea is to search on the letter **X** and use each **X** as a starting point, where we count **XMAS** occurrences in each of the 8 different directions. Let's create a function `count-xmases-at` that counts the number of **XMAS**-es starting from a location. Then simply call that for each **X** found in the grid and sum.

We're going to need to define two helper functions:

`char-locations` Returning all the locations of given character in the grid;

`is-xmas?` A function that takes the `grid`, a `start` coordinate and a `direction`.

It returns `true` if "XMAS" occurs in the `grid` from `start` in the given `direction`.

Since we're working in a grid, let's make `char-locations` return a 2d vector like `[0 0]` to denote the coordinates in the grid. While we're at it, let's define those directions like so:

```
12 (ns aoc.4)
13
14 (def directions
15   [[-1 0] ; Up
16    [ 1 0] ; Down
17    [ 0 -1] ; Left
18    [ 0 1] ; Right
19    [-1 -1] ; NW
20    [-1 1] ; NR
21    [ 1 -1] ; SW
22    [ 1 1]] ; SE
```

Which is a nice format to have, because you can then do something like this:

```
(defn move [start direction]
  (let [[x y] start
        [x' y'] direction]
    [(+ x x') (+ y y')]))

(let [start [4 5]
      direction [-1 0]] ;; up
  (println (move start direction)))

[3 5]
```

Which is really nice! So, let's define the final functions necessary for Part One:

```
23 (defn char-locations [grid x]
24   (mapcat
25     (fn [row string]
26       (keep-indexed (fn [col char] (when (= char x) [row col]))
27                     string))
28     (range)
29     grid))
30
31 (defn is-xmas? "Does the grid `grid` contain the string \"XMAS\",
32               starting at `start` and going in `direction`?"
33   [grid start direction]
34   (let [[x y] start
35         [x' y'] direction]
36     (loop [row x
37            col y
38            chars (seq "XMAS")]
39       (if (empty? chars)
40         true
41         (if (or (neg? row) (neg? col)
42                 (>= row (count grid)) (>= col (count (nth grid row)))
43                 (not= (get-in grid [row col]) (first chars)))
44           false
45           (recur (+ row x')
46                  (+ col y')
47                  (rest chars)))))))
```

Now we can tie everything together. `is-xmas?` returns true if the grid contains the word "XMAS" in a given direction. After we've found all X characters, we can count the amount of XMAS-es connected to it by counting all direction for which `is-xmas?` returns true.

```

48 (defn count-xmas-at [grid start directions]
49   (count (filter #(is-xmas? grid start %) directions)))
50
51 (defn p1 [input]
52   (let [grid (aoc.util/string-as-lines input)
53         xs (char-locations grid \X)]
54     (->> xs
55       (map #(count-xmas-at grid % directions))
56       (reduce +))))
57
58 (assert (= 18 (p1 example)))
59 (def input (slurp "inputs/4"))
60 (p1 input)

```

2447

4.2 Part 2

Ah, it seems the Elf thinks we're idiots because they use letters more literally. We don't need to find the string **XMAS**, we need to find the string **MAS** in an **X**, like so!

```

M.S
.A.
M.S

```

We could have reused the functionality above to search for **MAS**-es, and then only count a **MAS** that has a nice diagonal partner sharing the **A**. However, I found that a bit tricky to reason about, so I've opted to search for all of the **A**-s in the text, and finding **MAS** strings diagonally from that **A**. If there are exactly two **MAS**-es, we know that we got an **X-MAS**.

Instead of `is-xmas?`, we now have `is-mas?`, checking from a middle **A** instead of a starting **X**. Note that we're only counting **X-MAS**-es, so only use diagonals:

```

61 (def diagonal-directions
62   [[-1 -1] ; NW
63    [-1  1] ; NE
64    [ 1 -1] ; SW
65    [ 1  1]] ; SE

```

`is-mas?` is now pretty trivial:

```

66 (defn is-mas? [lines middle direction]
67   (let [[x y] middle
68         [x' y'] direction]
69     (and (= \M (get-in lines [(+ x x') (+ y y')]))
70          (= \S (get-in lines [(- x x') (- y y')])))))

```

And `count-mases-at` is virtually identical to `count-xmases-at` from Part One.

```
71 (defn count-mases-at [lines middle directions]
72   (count (filter #(is-mas? lines middle %) directions)))
73
74 (defn p2 [input]
75   (let [lines (aoc.util/string-as-lines input)
76         as (char-locations lines \A)]
77     (-> as
78         (map #(count-mases-at lines % diagonal-directions))
79         (filter #(= % 2))
80         (count))))
81
82 (assert (= 9 (p2 example)))
83 (p2 input)
```

1868